

8 Chapter: External Direct Products

In this chapter you will again need the file “orderFrequency”. You will need to read this file into GAP in order to do the exercises in this section. The command to form external direct products is `DirectProduct`. For example:

```
gap> C4:= CyclicGroup(IsPermGroup,4);
Group([ (1,2,3,4) ])
gap> S3:=SymmetricGroup(3);
Sym([ 1 .. 3 ])
gap> D:= DirectProduct(S3,C4);
Group([ (1,2,3), (1,2), (4,5,6,7) ])
gap> Size(D);
24
gap> Read("orderFrequency");
gap> orderFrequency(D);
[Order of element, Number of that order]=[ [ 1, 1 ], [ 2, 7 ], [ 3, 2 ],
[ 4, 8 ], [ 6, 2 ], [ 12, 4 ] ]
```

The first command above assigns the name $C4$ to the cyclic group of order 4 (generated by $(1,2,3,4)$). The third command forms the direct product $S3 \oplus C4$. The next two lines tell us that $S3 \oplus C4$ has 24 elements. (The order of the external direct product of two finite groups G_1 and G_2 is $|G_1||G_2|$.) From the last output above we see that $S3 \oplus C4$ has one element of order 1, seven elements of order 2, two elements of order 3, eight elements of order 4, two elements of order 6, and four elements of order 12.

Exercises

8.1 Find the number of elements of order 5 in $\mathbf{Z}_{25} \oplus \mathbf{Z}_5$. [Gallian, Chapter 8, Example 4]

8.2 Find the number of cyclic **subgroups** of order 10 in $\mathbf{Z}_{100} \oplus \mathbf{Z}_{25}$. (Hint: First find the number of elements of order 10. How many elements of order 10 are in a cyclic subgroup of order 10? Do any of these cyclic subgroups have an element of order 10 in common?) [Gallian, Chapter 8, Example 5]

8.3 **By hand** find the number of elements of each order in $D_{10} \oplus \mathbf{Z}_2$.

8.4 Check your answer to Exercise 8.3 using `orderFrequency`.

8.5 Use `orderFrequency` to find the number of elements of each order in $D_5 \oplus \mathbf{Z}_4$.

8.6 Are $D_{10} \oplus \mathbf{Z}_2$ and $D_5 \oplus \mathbf{Z}_4$ isomorphic? Why or why not?

8.7 **By hand** find the number of elements of each order in D_{20} .

8.8 Check your answer to Exercise 8.7 using `orderFrequency`. Is D_{20} isomorphic to either $D_{10} \oplus \mathbf{Z}_2$

or $D_5 \oplus Z_4$?

8.9 Find 4 nonisomorphic groups of order 40. How many nonisomorphic groups of order 40 can you find?

The command `AllSmallGroups(n)` gives a list of all groups of order n . (Type `?AllSmallGroups` while in `GAP` to see the limitations on the integers n that can be used in this command.) For example the following is a list of all groups of order 20:

```
gap> Gorder20:= AllSmallGroups(20);
[ <pc group of size 20 with 3 generators>,
  <pc group of size 20 with 3 generators>,
  <pc group of size 20 with 3 generators>,
  <pc group of size 20 with 3 generators>,
  <pc group of size 20 with 3 generators> ]
```

The output does not appear to be useful. But we see there are five groups of order 20. We can refer to each of these groups in the list. For example, `Gorder20[1]` refers to the first group of order 20 listed above. We can now explore properties of each of these five groups:

```
gap> Read("orderFrequency");
gap> orderFrequency(Gorder20[1]);
[Order of element, Number of that order]=[ [ 1, 1 ], [ 2, 1 ], [ 4, 10 ],
[ 5, 4 ], [ 10, 4 ] ]
gap> IsAbelian(Gorder20[1]);
false
gap> d10:= DihedralGroup(IsPermGroup,20);
Group([ (1,2,3,4,5,6,7,8,9,10), (2,10)(3,9)(4,8)(5,7) ])
gap> IsomorphismGroups(d10, Gorder20[1]);
fail
gap> IsomorphismGroups(d10, Gorder20[4]);
[ (1,2,3,4,5,6,7,8,9,10), (2,10)(3,9)(4,8)(5,7) ] -> [ f2*f3^2, f1*f2*f3^4 ]
```

From the output of the last and next to last above commands we see that the first group listed is not isomorphic to D_{10} but the fourth group listed is isomorphic to D_{10} .

By looping through a list we can actually get the “orderFrequency”, for example, of each of these groups of order 20:

```
gap> List(AllSmallGroups(20) , x -> orderFrequency(x));
[ [ [ 1, 1 ], [ 2, 1 ], [ 4, 10 ], [ 5, 4 ], [ 10, 4 ] ],
  [ [ 1, 1 ], [ 2, 1 ], [ 4, 2 ], [ 5, 4 ], [ 10, 4 ], [ 20, 8 ] ],
  [ [ 1, 1 ], [ 2, 5 ], [ 4, 10 ], [ 5, 4 ] ],
  [ [ 1, 1 ], [ 2, 11 ], [ 5, 4 ], [ 10, 4 ] ],
  [ [ 1, 1 ], [ 2, 3 ], [ 5, 4 ], [ 10, 12 ] ] ]
```

In GAP a matrix can be entered as a list of row vectors. For example, the matrix

$$M = \begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \end{bmatrix}$$

is entered by typing

```
gap> M:= [ [1,2,3], [4,5,6], [7,8,9]];
      [ [ 1, 2, 3 ], [ 4, 5, 6 ], [ 7, 8, 9 ] ]
```

If you prefer to exhibit the matrix M as a 3 by 3 array use the command `PrintArray`:

```
gap> PrintArray(M);
      [ [ 1, 2, 3 ],
        [ 4, 5, 6 ],
        [ 7, 8, 9 ] ]
```

8.10 Let $G = \mathbf{Z}_3 \oplus \mathbf{Z}_3 \oplus \mathbf{Z}_3$ and let H be the subgroup of $\mathrm{SL}(3, \mathbf{Z}_3)$ consisting of

$$H = \left\{ \begin{bmatrix} 1 & a & b \\ 0 & 1 & c \\ 0 & 0 & 1 \end{bmatrix} \mid a, b, c \in \mathbf{Z}_3 \right\}.$$

Determine the number of elements of each order in G and H . Are G and H isomorphic? (This exercise shows that two groups with the same number of elements of each order need not be isomorphic.) [Gallian, Supplementary Exercises for Chapter 5-8, Exercise 5] Hint: Since H is generated by

$$\begin{bmatrix} 1 & 1 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 1 & 0 & 1 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}, \text{ and } \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 1 \\ 0 & 0 & 1 \end{bmatrix}$$

the following sets up this group H :

```
gap> z:= Z(3);;
gap> a:= [[z^0, z^0, 0*z], [0*z, z^0, 0*z], [0*z, 0*z, z^0]];;
gap> b:= [[z^0, 0*z, z^0], [0*z, z^0, 0*z], [0*z, 0*z, z^0]];;
gap> c:= [[z^0, 0*z, 0*z], [0*z, z^0, z^0], [0*z, 0*z, z^0]];;
gap> H:= Subgroup(SL(3,3),[a,b,c]);
<matrix group with 3 generators>
```